BRL

MEMORANDUM REPORT NO. 2711

NEWT: A COMPUTER CODE FOR CALCULATING THE KINEMATIC VARIABLES OF A PROJECTILE SUBJECT TO A SECOND ORDER DRAG LAW THAT IS PENETRATING AN ELASTIC-PLASTIC MEDIUM

William B. Beverly

December 1976

USA BALLISTIC RESEARCH LABORATORIES
ABERDEEN PROVING GROUND, MARYLAND

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE

READ INSTRUCTIONS BEFORE COMPLETING FORM

1. REPORT NUMBER: BRL MEMORANDUM REPORT NO. 2711

2. GOVT ACCESSION NO.:

3. RECIPIENT'S CATALOG NUMBER: (9)

4. TITLE (and Subtitle): (6) NEWT: A Computer Code for Calculating the Kinematic Variables of a Projectile Subject to a Second Order Drag Law That is Penetrating an Elastic-Plastic Medium.

5. TYPE OF REPORT & PERIOD COVERED: Final rept.

6. PERFORMING ORG. REPORT NUMBER:

7. AUTHOR(s): (10) William B. Beverly

8. CONTRACT OR GRANT NUMBER(s):

9. PERFORMING ORGANIZATION NAME AND ADDRESS:
US Army Ballistic Research Laboratories
Aberdeen Proving Ground, MD 21005

10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS: (16) 1W662618AH80

11. CONTROLLING OFFICE NAME AND ADDRESS:
USA Materiel Development & Readiness Command
5001 Eisenhower Avenue
Alexandria, VA 22333

12. REPORT DATE: (11) DEC[redacted]76

13. NUMBER OF PAGES: 35

14. MONITORING AGENCY NAME & ADDRESS (if different from Controlling Office): (12) 29p.

15. SECURITY CLASS. (of this report): UNCLASSIFIED

15a. DECLASSIFICATION/DOWNGRADING SCHEDULE: N/A

16. DISTRIBUTION STATEMENT (of this Report):

Approved for public release; distribution unlimited.

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report):

(14) BRL-MR-2711

18. SUPPLEMENTARY NOTES:

19. KEY WORDS (Continue on reverse side if necessary and identify by block number):

Ballistic Penetration
Elastic-Plastic Medium
Differential Equation
Second Order Drag Force

20. ABSTRACT (Continue on reverse side if necessary and identify by block number):

The motion of a projectile in an elastic-plastic medium can often be adequately described by a second order drag law. The differential equations describing this motion have been solved for a projectile of impacting velocity V_o and frontal area πR^2 and a computer program NEWT has been written in the BASIC language to calculate the penetration, velocity, and energy deposition of the projectile at subsequent times. The code was written to describe projectiles in tissue simulant but it can be used to predict any motion described by a second-order drag law.

(continued)

sub 0

pi R squared

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 65 IS OBSOLETE

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

050750

1B

The differential equations and their solutions are discussed in detail. A flow chart and code listing are given. A description of the user-supplied input is given. Representative test problems are solved and their solutions are discussed.

TABLE OF CONTENTS

I. INTRODUCTION

A general empirical expression for the motion of a projectile of mass m and velocity V obeying a second order drag law has been proposed that has the form[1]

$$F = m\frac{dV}{dT} = - A'V^2 - B'V - C' . \qquad (1)$$

The coefficients A', B', and C' are quantities that can be either evaluated experimentally from penetration-time data or calculated using basic quantities. The terms on the right side of Eq. 1 can be considered to be the forces contributing to three components of energy deposition. These three components or modes of energy deposition are: (1) the acceleration of target material adjacent to the projectile, (2) the effect of frictional forces, and (3) the cohesive strength of the target material respectively. A comprehensive bibliography of earlier work using versions of Eq. 1 where one or two of the coefficients were set to zero is given by Goldsmith[2] on p.298.

Dienes and Welch[3] define the coefficients in terms of basic physical quantities. Their definitions are:

$$A' = \frac{\pi}{2} \rho C_D r^2 \qquad (1A)$$

$$B' = 6\pi r \eta \qquad (1B)$$

$$C' = \pi r^2 \sigma \qquad (1C)$$

where ρ is the target density and πr^2 is the frontal area of the projectile. The quantity C_D is the drag coefficient, η is the dynamic viscosity, and σ is the normal stress.

The rate of work by the projectile as it penetrates the target is equal to its rate of loss of kinetic energy and is given by

$$F \cdot dS = d(\frac{1}{2} mV^2) \qquad (2A)$$

---

1. *W.A. Allen, E.B. Mayfield, and H.L. Morrison, "Dynamics of a Projectile Penetrating Sand," Journal of Applied Physics, 28, 1957.*

2. *Weiner Goldsmith, "Impact," Edward Arnold (Publishers) Ltd., London, England, 1960.*

3. *J.K. Dienes and J.E. Welch, "Parameter Study of Rigid Body Penetration," Systems, Science and Software Report No. SSS-R-75-2521, 1974.*

This leads in the limit to

$$F = mV\frac{dV}{dS} = - A'V^2 - B'V - C' \quad . \qquad (2B)$$

Eqs. 1 and 2B are solved for an impacting velocity V_o at time zero to obtain analytical relations for the velocity and penetration as functions of time. The solutions are discussed in the next section.

A computer code NEWT was written, using the BASIC programming language, to calculate the penetration and velocity values for a given impacting velocity and a given set of coefficients at times subsequent to impact. The energy deposited (both total and the components contributed by each force component) along the projectile trajectory are also calculated.

A listing of the code is given in Appendix B and a description of the code that includes a flow chart is given in Section IV. A description of the code input, to be supplied by the user, is given in Section V. Sample problems that demonstrate the differing effect of each coefficient on the projectile penetration are discussed in Appendix A.

The code has been run successfully on the WANG 2200 calculator. Results and identifying captions are printed by the calculator-controlled typewriter. Approximately 5000 words of memory are needed to execute the program.

II. CALCULATIONAL PROCEDURE

Equation 1 may be expressed in integral form as

$$\int_{V_o}^{V} \frac{dV}{AV^2 + BV + C} = - \int_{o}^{T} dT = -T \quad . \qquad (3)$$

where V_o and V are the velocities at time zero and T respectively. The constant m has been absorbed into the coefficients yielding the unprimed set. The integrals can be solved in closed form where the solution to the leftmost integral may assume one of the three forms dependent upon the value of the discriminator R1 where

$$R1 = B^2 - 4AC \quad . \qquad (4)$$

The three forms are:

1. $R1 > 0$,

$$T = \left| \frac{1}{R} \ln \frac{2AV + B - R}{2AV + B + R} \right|_{V}^{V_o} , \tag{5A}$$

2. $R1 < 0$,

$$T = \left| \frac{2}{R} \tan^{-1} \frac{2AV + B}{R} \right|_{V}^{V_o} , \tag{5B}$$

3. $R1 = 0$,

$$T = \left| \frac{-2}{2AV + B} \right|_{V}^{V_o} . \tag{5C}$$

The quantity R is the positive real value of

$$R = \sqrt{\pm R1} . \tag{6}$$

Eq. 2B in integral form can be expressed as

$$\int_{V}^{V_o} \frac{VdV}{AV^2 + BV + C} = - \int_{S}^{o} dS = S . \tag{7}$$

The solution to this integral is

$$S = \left| \frac{1}{2A} \ln(AV^2 + BV + C) \right|_{V}^{V_o} - \frac{B}{2A} \int_{V}^{V_o} \frac{dV}{AV^2 + BV + C} . \tag{8}$$

It should be noted that the rightmost integral has the same form as that derived in Eq. 3. This permits Eq. 8 to be expressed as

$$S = \left| \frac{1}{2A} \ln (AV^2 + BV + C) \right|_{V}^{V_o} - \frac{BT}{2A} \tag{9}$$

or

$$S = \frac{1}{2A} \left[\ln \frac{AV_o^2 + BV_o + C}{AV^2 + BV + C} - BT \right] . \tag{10}$$

Returning to Eq. 3, its solution for positive values of R1 is given by

$$T = \left| \frac{1}{R} \ln \frac{2AV + B - R}{2AV + B + R} \right|_{V}^{V_o} \qquad (11)$$

which expands to

$$T = \frac{1}{R} \ln \left(\frac{1}{G} \frac{2AV + B + R}{2AV + B - R} \right) \qquad (12A)$$

where

$$G = \frac{2AV_o + B + R}{2AV_o + B - R} . \qquad (12B)$$

The term G will always be positive. The velocity is given explicitly by

$$V = \frac{R + B + Ge^{RT}(R - B)}{2A(Ge^{RT} - 1)} . \qquad (13)$$

That relation, as expected, will give a value of V_o to the velocity when T is zero.

The solution to Eq. 3 for negative values of R1 is given by

$$T = \left| \frac{2}{R} \tan^{-1} \frac{2AV + B}{R} \right|_{V}^{V_o} \qquad (14)$$

which expands to

$$T = \frac{2}{R} \left(\tan^{-1} \frac{2AV_o + B}{R} - \tan^{-1} \frac{2AV + B}{R} \right) . \qquad (15)$$

The velocity is given explicitly by

$$V = \frac{R}{2A} \left[\tan(\tan^{-1} \frac{2AV_o + B}{R} - \frac{RT}{2}) - \frac{B}{R} \right] . \qquad (16)$$

The solution to Eq. 3 for a zero value of R1 is given by

$$T = \left| \frac{-2}{2AV + B} \right|_{V}^{V_o} \qquad (17)$$

which expands to

$$T = 2 \left[\frac{1}{2AV + B} - \frac{1}{2AV_o + B} \right] . \qquad (18)$$

The velocity is given explicitly by

$$V = \frac{4AV_o - BT(2AV_o + B)}{2A[2 + T(2AV_o + B)]} . \qquad (19)$$

The preceding solutions are valid only for positive, non-zero values of the coefficient A. The integral in Eq. 3 was solved for zero values of A and positive, non-zero values of B yielding

$$V = \frac{1}{B}\left[e^{-BT}(BV_o + C) - C\right] . \quad (20)$$

Similarly, Eq. 8 was solved for the penetration yielding

$$S = \frac{1}{B}\left[V_o - V + \frac{C}{B}\ln\frac{BV + C}{BV_o + C}\right] . \quad (21)$$

None of the preceding solutions are valid if both A and B are zero. For this case the solution to Eq. 3 was found to be

$$V = V_o - CT . \quad (22)$$

The solution for the penetration was

$$S = \frac{1}{2C}\left[V_o{}^2 - V^2\right] . \quad (23)$$

III. CODE DESCRIPTION

The code NEWT was designed so that it would check the input values of A, B, and C, evaluate the discriminator if necessary, choose the proper set of equations to use, and calculate the penetration, velocity, and energy deposition rates at the chosen times. The input and the calculated results are then printed in tabular form along with their captions.

A flow chart of NEWT is displayed in Figure 1. The important variables and arrays are listed and briefly described in Table I.



Figure 1. NEWT Flow Chart

Table I. Variables and Arrays

Variable	Description
A3	The coefficient A in Equation 1
B3	The coefficient B in Equation 1
C3	The coefficient C in Equation 1
A	$A = \overline{A3}\ M/0.5\pi R^2$
B	$B = \overline{B3}\ M/6\pi R$
C	$C = C3\ M/\pi R^2$
R	Effective impacting radius of Projectile
M	Mass of Projectile
V_o	Initial Velocity of Projectile
Q7	Calculation mode flag
	Q7 = 001, $A = B = 0,\ C > 0$
	Q7 = 010, $A = C = 0,\ B > 0$
	Q7 = 011, $A = C = 0,\ C > 0$
	Q7 = 100, $A > 0,\ B = C = 0$
	Q7 = 101, $A > 0,\ B = 0,\ C > 0$
	Q7 = 111, $A > 0,\ B > 0,\ C > 0$
F1	Calculation discriminator
	F1 = 1, $B^2 - 4AC > 0$
	F1 = 2, $B^2 - 4AC < 0$
	F1 = 3, $B^2 - 4AC = 0$
N1	Number of times
T	Times
T1(50)	Times array
V	Velocity
V1(50)	Velocity array
S	Penetration
S1(50)	Penetration array
R1	$R1 = \sqrt{\pm(B^2 - 4AC)}$

IV. INPUT

The user-supplied input needed to execute the program is listed in Table II. It is listed in the order by which it is requested by NEWT. CGS units should be used to conform to captions but they could be easily changed to conform to another system. It should be noted that the programming variables A, B, and C are not the coefficients A, B, and C. The coefficients A, B, and C correspond to the programming variables A3, B3, and C3.

Table II. Code-Generated Queries

Instruction	Description
1. Title =	Problem Title
2. A =	$A = \rho C_D$ (Eq. 1A)
3. B =	$B = \eta$(Eq. 1B)
4. C =	$C = \sigma$(Eq. 1C)
5. Radius =	Frontal Area = πr^2
6. Mass =	Mass = (gm)
7. Initial Velocity =	V_o = (cm/sec)
8. No Times =	No of times for which kinetic quantities are to be calculated
9. 1 for Read-in Times =	F2 = 0 Calculates times
10. Times Interval =	Time Interval (calculated times only)
11. Next Time =	Next Time (Read-in Times only)

V. CONCLUSIONS

This study, and the building of the Computer Code, were conducted at the Target Assessments Branch (TAB), Vulnerability/Lethality Division(VLD), Ballistic Research Laboratories (BRL) as part of an on-going effort to calculate the penetration of human tissue by bullets or fragments and to derive the resulting incapacitation probabilities. However, the code is not restricted to a tissue media, but can be used to calculate the penetration of any media that is described by a second order drag law. A magnetic tape copy of the code can be obtained by sending a blank computer tape cassette to the TAB.

Earlier work by Bruchey[4], in performing least square determination of drag coefficients for handgun bullets penetrating tissue simulant, has provided a validation of the model for non-zero values of the A- and C-coefficients and a zero value of the B-coefficients. It would therefore be expected that a better fit of experimental data to the model, as well as a better description of the energy deposition modes in the medium, could be obtained by optimizing all three coefficients instead of only the A-coefficients as was done in the preceding work. A computer code for providing this least squares fit of experimentally-determined penetration-time data to the model by optimizing all three coefficients is being developed by the TAB.[5]

The degree of human incapacitation (however this term is ultimately quantized and defined) will probably be the sum of the products formed by the volume of a particular tissue destroyed and its vulnerability index. The volume of tissue destroyed is approximated by the volume of the maximum temporary wound cavity (TWC). A future effort will be made to construct a wound cavity model that used the energy deposition rates to predict the size of the TWC. It is expected that the forth-coming study and the resulting Computer Code will provide a rapid and inexpensive means of calculating the volume of that cavity.

---

4. *William J.Bruchey, Jr., "Effective Drag Coefficients for Various Small Arms Projectiles in Dense Media," BRL Memorandum Report 2481, May 1975.* (AD #B004154L)

5. *William B. Beverly, "LSP: A Least Square Program for Calculating The Resistive Force Coefficients of a Projectile Subject to a Second Order Drag Law," to be published as a BRL Report.*

ACKNOWLEDGEMENT

The author wishes to acknowledge the contribution of Mr. Eugene T. Roecker whose suggestion greatly simplified the mathematical equations and permitted more efficient computer coding of NEWT.

APPENDIX A

SAMPLE PROBLEMS

APPENDIX A

SAMPLE PROBLEMS

Sample values A, B, and C were chosen so that the three components of the opposing force would be equal at the impacting velocity of 30,000 cm/sec. This choice of input values permits an easy analysis of the kinematic results for demonstrating the differing effects of the three force components. The penetration and velocity versus time are listed in Table A-I, the partial differential (in space) energy losses are listed in Table A-II, and the partial cumulative energy losses are listed in Table A-III.

Three more calculations were conducted in which A, B, and C in turn were assigned a non-zero value (that used in the first problem) while the remaining two coefficients were set to zero. The penetration and velocity versus time values calculated for these problems are listed in Tables A-IV - A-VI. The penetration and velocity versus time for all test problems are displayed in Figure A-1.

Table A-I. Kinematic Results (A>0, B>0, C>0)

```
TEST PROBLEM 1
MASS=   8.000E+00 RADIUS=   6.000E-01 INIT VEL= 3.000E+04
A=      1.200E+00 B=        1.800E+03 C=        5.399E+08
A3=     8.482E-02 B3=       2.544E+03 C3=       7.633E+07
```

KINEMATIC RESULTS

TIME(SEC)	PEN(CM)	VEL(CM/SEC)	DEL E/DEL S
0.0000E+00	0.0000E+00	3.0000E+04	2.2901E+08
1.0000E-05	2.8883E-01	2.7793E+04	2.1258E+08
2.0000E-05	5.5638E-01	2.5741E+04	1.9805E+08
3.0000E-05	8.0412E-01	2.3827E+04	1.8512E+08
4.0000E-05	1.0333E+00	2.2034E+04	1.7359E+08
5.0000E-05	1.2451E+00	2.0351E+04	1.6325E+08
6.0000E-05	1.4406E+00	1.8766E+04	1.5396E+08
7.0000E-05	1.6208E+00	1.7269E+04	1.4557E+08
8.0000E-05	1.7863E+00	1.5851E+04	1.3799E+08
9.0000E-05	1.9380E+00	1.4506E+04	1.3110E+08
1.0000E-04	2.0767E+00	1.3227E+04	1.2484E+08
1.1000E-04	2.2028E+00	1.2008E+04	1.1912E+08
1.2000E-04	2.3170E+00	1.0843E+04	1.1390E+08
1.3000E-04	2.4198E+00	9.7287E+03	1.0912E+08
1.4000E-04	2.5117E+00	8.6597E+03	1.0473E+08
1.5000E-04	2.5932E+00	7.6328E+03	1.0070E+08
1.6000E-04	2.6645E+00	6.6446E+03	9.6991E+07
1.7000E-04	2.7262E+00	5.6920E+03	9.3570E+07
1.8000E-04	2.7785E+00	4.7723E+03	9.0413E+07
1.9000E-04	2.8217E+00	3.8829E+03	8.7497E+07
2.0000E-04	2.8562E+00	3.0216E+03	8.4801E+07
2.1000E-04	2.8822E+00	2.1862E+03	8.2306E+07
2.2000E-04	2.9000E+00	1.3749E+03	7.9996E+07
2.3000E-04	2.9098E+00	5.8577E+02	7.7857E+07

Table A-II. Differential Energy Deposition (A>0, B>0, C>0)

DIFFERENTIAL ENERGY LOSS(ERGS/CM)			
TIME(SEC)	A3*V*V	B3*V	C3
0.0000E+00	7.6340E+07	7.6340E+07	7.6337E+07
1.0000E-05	6.5524E+07	7.0726E+07	7.6337E+07
2.0000E-05	5.6207E+07	6.5505E+07	7.6337E+07
3.0000E-05	4.8157E+07	6.0633E+07	7.6337E+07
4.0000E-05	4.1184E+07	5.6071E+07	7.6337E+07
5.0000E-05	3.5131E+07	5.1788E+07	7.6337E+07
6.0000E-05	2.9871E+07	4.7753E+07	7.6337E+07
7.0000E-05	2.5296E+07	4.3944E+07	7.6337E+07
8.0000E-05	2.1314E+07	4.0338E+07	7.6337E+07
9.0000E-05	1.7851E+07	3.6915E+07	7.6337E+07
1.0000E-04	1.4841E+07	3.3660E+07	7.6337E+07
1.1000E-04	1.2231E+07	3.0557E+07	7.6337E+07
1.2000E-04	9.9737E+06	2.7593E+07	7.6337E+07
1.3000E-04	8.0283E+06	2.4756E+07	7.6337E+07
1.4000E-04	6.3610E+06	2.2036E+07	7.6337E+07
1.5000E-04	4.9418E+06	1.9423E+07	7.6337E+07
1.6000E-04	3.7450E+06	1.6908E+07	7.6337E+07
1.7000E-04	2.7482E+06	1.4484E+07	7.6337E+07
1.8000E-04	1.9318E+06	1.2144E+07	7.6337E+07
1.9000E-04	1.2789E+06	9.8809E+06	7.6337E+07
2.0000E-04	7.7447E+05	7.6891E+06	7.6337E+07
2.1000E-04	4.0544E+05	5.5634E+06	7.6337E+07
2.2000E-04	1.6034E+05	3.4987E+06	7.6337E+07
2.3000E-04	2.9105E+04	1.4906E+06	7.6337E+07

Table A-III. Cumulative Energy Deposition (A>0, B>0, C>0)

CUMULATIVE ENERGY LOSS(ERGS)			
TIME(SEC)	INT(A3*V*V)DS	INT(B3*V)DS	INT(C3)DS
0.0000E+00	0.0000E+00	0.0000E+00	0.0000E+00
1.0000E-05	1.6390E+08	1.6991E+08	1.7639E+08
2.0000E-05	2.9418E+08	3.1570E+08	3.3978E+08
3.0000E-05	3.9760E+08	4.4070E+08	4.9108E+08
4.0000E-05	4.7951E+08	5.4770E+08	6.3106E+08
5.0000E-05	5.4418E+08	6.3910E+08	7.6043E+08
6.0000E-05	5.9502E+08	7.1695E+08	8.7983E+08
7.0000E-05	6.3476E+08	7.8301E+08	9.8982E+08
8.0000E-05	6.6562E+08	8.3882E+08	1.0909E+09
9.0000E-05	6.8940E+08	8.8571E+08	1.1835E+09
1.0000E-04	7.0752E+08	9.2484E+08	1.2682E+09
1.1000E-04	7.2118E+08	9.5724E+08	1.3452E+09
1.2000E-04	7.3133E+08	9.8381E+08	1.4150E+09
1.3000E-04	7.3873E+08	1.0053E+09	1.4778E+09
1.4000E-04	7.4402E+08	1.0225E+09	1.5339E+09
1.5000E-04	7.4770E+08	1.0360E+09	1.5836E+09
1.6000E-04	7.5018E+08	1.0464E+09	1.6272E+09
1.7000E-04	7.5178E+08	1.0541E+09	1.6649E+09
1.8000E-04	7.5276E+08	1.0597E+09	1.6968E+09
1.9000E-04	7.5332E+08	1.0635E+09	1.7232E+09
2.0000E-04	7.5360E+08	1.0659E+09	1.7443E+09
2.1000E-04	7.5372E+08	1.0673E+09	1.7602E+09
2.2000E-04	7.5376E+08	1.0679E+09	1.7710E+09
2.3000E-04	7.5377E+08	1.0681E+09	1.7770E+09

Table A-IV. Kinematic Results (A>0, B=0, C=0)

TEST PROBLEM 2

MASS=	8.000E+00	RADIUS=	6.000E-01	INIT VEL=	3.000E+04
A=	1.200E+00	B=	0.000E+00	C=	0.000E+00
A3=	8.482E-02	B3=	0.000E+00	C3=	0.000E+00

KINEMATIC RESULTS

TIME(SEC)	PEN(CM)	VEL(CM/SEC)	DEL E/DEL S
0.0000E+00	0.0000E+00	3.0000E+04	7.6340E+07
1.0000E-05	2.9624E-01	2.9255E+04	7.2598E+07
2.0000E-05	5.8523E-01	2.8547E+04	6.9125E+07
3.0000E-05	8.6730E-01	2.7872E+04	6.5895E+07
4.0000E-05	1.1427E+00	2.7228E+04	6.2886E+07
5.0000E-05	1.4119E+00	2.6613E+04	6.0079E+07
6.0000E-05	1.6751E+00	2.6026E+04	5.7456E+07
7.0000E-05	1.9325E+00	2.5464E+04	5.5001E+07
8.0000E-05	2.1845E+00	2.4925E+04	5.2699E+07
9.0000E-05	2.4311E+00	2.4409E+04	5.0540E+07
1.0000E-04	2.6727E+00	2.3914E+04	4.8510E+07
1.1000E-04	2.9095E+00	2.3439E+04	4.6600E+07
1.2000E-04	3.1416E+00	2.2982E+04	4.4801E+07
1.3000E-04	3.3692E+00	2.2542E+04	4.3104E+07
1.4000E-04	3.5925E+00	2.2119E+04	4.1502E+07
1.5000E-04	3.8116E+00	2.1712E+04	3.9987E+07
1.6000E-04	4.0268E+00	2.1319E+04	3.8554E+07
1.7000E-04	4.2381E+00	2.0940E+04	3.7196E+07
1.8000E-04	4.4456E+00	2.0575E+04	3.5909E+07
1.9000E-04	4.6496E+00	2.0222E+04	3.4688E+07
2.0000E-04	4.8501E+00	1.9881E+04	3.3528E+07
2.1000E-04	5.0473E+00	1.9551E+04	3.2425E+07
2.2000E-04	5.2412E+00	1.9232E+04	3.1376E+07
2.3000E-04	5.4320E+00	1.8924E+04	3.0377E+07

Table A-V. Kinematic Results (B>0, A=C=0)

TEST PROBLEM 3

MASS=	8.000E+00	RADIUS=	6.000E-01	INIT VEL=	3.000E+04
A=	0.000E+00	B=	1.800E+03	C=	0.000E+00
A3=	0.000E+00	B3=	2.544E+03	C3=	0.000E+00

KINEMATIC RESULTS

TIME(SEC)	PEN(CM)	VEL(CM/SEC)	DEL E/DEL S
0.0000E+00	0.0000E+00	3.0000E+04	7.6340E+07
1.0000E-05	2.9621E-01	2.9246E+04	7.4422E+07
2.0000E-05	5.8498E-01	2.8511E+04	7.2552E+07
3.0000E-05	8.6650E-01	2.7795E+04	7.0729E+07
4.0000E-05	1.1409E+00	2.7096E+04	6.8952E+07
5.0000E-05	1.4084E+00	2.6415E+04	6.7220E+07
6.0000E-05	1.6693E+00	2.5752E+04	6.5531E+07
7.0000E-05	1.9235E+00	2.5105E+04	6.3884E+07
8.0000E-05	2.1714E+00	2.4474E+04	6.2279E+07
9.0000E-05	2.4131E+00	2.3859E+04	6.0714E+07
1.0000E-04	2.6487E+00	2.3259E+04	5.9189E+07
1.1000E-04	2.8783E+00	2.2675E+04	5.7701E+07
1.2000E-04	3.1022E+00	2.2105E+04	5.6252E+07
1.3000E-04	3.3205E+00	2.1550E+04	5.4838E+07
1.4000E-04	3.5333E+00	2.1008E+04	5.3460E+07
1.5000E-04	3.7407E+00	2.0480E+04	5.2117E+07
1.6000E-04	3.9429E+00	1.9966E+04	5.0808E+07
1.7000E-04	4.1401E+00	1.9464E+04	4.9531E+07
1.8000E-04	4.3323E+00	1.8975E+04	4.8286E+07
1.9000E-04	4.5196E+00	1.8498E+04	4.7073E+07
2.0000E-04	4.7023E+00	1.8034E+04	4.5890E+07
2.1000E-04	4.8804E+00	1.7580E+04	4.4737E+07
2.2000E-04	5.0539E+00	1.7139E+04	4.3613E+07
2.3000E-04	5.2232E+00	1.6708E+04	4.2518E+07

Table A-VI. Kinematic Results (C>0, A=B=0)

TEST PROBLEM 4

MASS=	8.000E+00	RADIUS=	6.000E-01	INIT VEL=	3.000E+04
A=	0.000E+00	B=	0.000E+00	C=	5.399E+03
A3=	0.000E+00	B3=	0.000E+00	C3=	7.633E+07

KINEMATIC RESULTS

TIME(SEC)	PEN(CM)	VEL(CM/SEC)	DEL E/DEL S
0.0000E+00	0.0000E+00	3.0000E+04	7.6337E+07
1.0000E-05	2.9618E-01	2.9236E+04	7.6337E+07
2.0000E-05	5.8473E-01	2.8473E+04	7.6337E+07
3.0000E-05	8.6564E-01	2.7709E+04	7.6337E+07
4.0000E-05	1.1389E+00	2.6946E+04	7.6337E+07
5.0000E-05	1.4045E+00	2.6183E+04	7.6337E+07
6.0000E-05	1.6625E+00	2.5419E+04	7.6337E+07
7.0000E-05	1.9129E+00	2.4656E+04	7.6337E+07
8.0000E-05	2.1557E+00	2.3892E+04	7.6337E+07
9.0000E-05	2.3908E+00	2.3129E+04	7.6337E+07
1.0000E-04	2.6183E+00	2.2366E+04	7.6337E+07
1.1000E-04	2.8381E+00	2.1602E+04	7.6337E+07
1.2000E-04	3.0503E+00	2.0839E+04	7.6337E+07
1.3000E-04	3.2549E+00	2.0076E+04	7.6337E+07
1.4000E-04	3.4518E+00	1.9312E+04	7.6337E+07
1.5000E-04	3.6411E+00	1.8549E+04	7.6337E+07
1.6000E-04	3.8228E+00	1.7785E+04	7.6337E+07
1.7000E-04	3.9969E+00	1.7022E+04	7.6337E+07
1.8000E-04	4.1633E+00	1.6259E+04	7.6337E+07
1.9000E-04	4.3221E+00	1.5495E+04	7.6337E+07
2.0000E-04	4.4732E+00	1.4732E+04	7.6337E+07
2.1000E-04	4.6167E+00	1.3969E+04	7.6337E+07
2.2000E-04	4.7526E+00	1.3205E+04	7.6337E+07
2.3000E-04	4.8808E+00	1.2442E+04	7.6337E+07



Figure A-1. Results of Test Problems

APPENDIX B

NEWT PROGRAM LISTING

APPENDIX B

NEWT PROGRAM LISTING

```
1REM NEWT3
2REM F=-A*V*V-B*V-C
3INPUT "TITLE",Z2$,Z3$,Z4$,Z5$
4SELECT LIST 211
5SELECT PRINT 211
6SELECT R
10DIM T1(50),V1(50),S1(50)
15INPUT "A=",A
20INPUT "B=",B
25INPUT "C=",C
30INPUT "RADIUS=",R
35INPUT "MASS=",M
40INPUT "INITIAL VELOCITY=",V0
45INPUT "NO. TIMES=",N1
50INPUT "1 FOR READ-IN TIMES",F2
55IF F2=0THEN 75
60FOR I=1TO N1
65INPUT "NEXT TIME=",T1(I)
70NEXT I
75A3=#PI*0.5*R*R*A/M
80B3=6.0*#PI*R*B/M
85C3=#PI*R*R*C/M
90IF F2]0THEN 125
95T=0.0
100INPUT "TIME INTERVAL=",Z1
105FOR I=1TO N1
110T1(I)=T
115T=T+Z1
120NEXT I
125IF A]0THEN 134
127IF B]0THEN 130
128Q7=001: GOTO 220
130IF C]0THEN 132
131Q7=010: GOTO 220
132Q7=011: GOTO 220
134IF B]0THEN 138
135IF C]0THEN 137
136Q7=100: GOTO 154
137Q7=101: GOTO 154
138IF C]0THEN 140
```

```
139Q7=110: GOTO 154
140Q7=111
154REM CALCULATE B*B-4*A*C FLAG
155F1=B3*B3-4.0*A3*C3
160IF F1[0.0THEN 175
165IF F1]0.0THEN 180
170F1=3:GOTO 220
175F1=2:GOTO 220
180F1=1
220REM  PRINT INPUT AND CAPTIONS
222PRINT Z2$,Z3$,Z4$,Z5$
223Z2$="MASS="
224Z3$="RADIUS="
225Z4$="INIT VEL="
226PRINTUSING 227,Z2$,M,Z3$,R,Z4$,V0
227%######### #.###!!!! ######### #.###!!!! ######### #.###!!!!
228Z2$="A="
229Z3$="B="
230Z4$="C="
231PRINTUSING 227,Z2$,A,Z3$,B,Z4$,C
232Z2$="A3="
233Z3$="B3="
234Z4$="C3="
235PRINTUSING 227,Z2$,A3,Z3$,B3,Z4$,C3
245GOSUB '01
246PRINT "KINEMATIC RESULTS"
250Z1$="TIME(SEC)":Z2$="PEN(CM)":Z3$="VEL(CM/SEC)"
251Z4$="DEL E/DEL S"
255PRINTUSING 260,Z1$,Z2$,Z3$,Z4$
260%#########       #######        ###########   ###########
265GOSUB '01
270FOR I=1TO N1
275IF Q7]1THEN 295
280GOSUB '11(T1(I))
285GOSUB '21
290GOTO 340
295IF Q7]11THEN 315
300GOSUB '13(T1(I))
305GOSUB '23
310GOTO 340
315IF Q7]111THEN 335
320GOSUB '16(T1(I))
325GOSUB '26
330GOTO 340
335A9=2:GOTO 430
340NEXT I
342IF T1(1)]0.0THEN 345
343S1(1)=0.0
345FOR I=1TO N1
346IF V1(I)[0THEN 352
```

```
350PRINTUSING 355,T1(I),S1(I),V1(I),A3*V1(I)!2+B3*V1(I)+C3
351GOTO 360
352PRINT "NEGATIVE VELOCITY"
353N1=I-1
354GOTO 365
355%#.####!!!!     #.####!!!!     #.####!!!!     #.####!!!!
360NEXT I
365GOSUB '01
367FOR I=1TO 5
368PRINT
369NEXT I
370GOSUB '01
371Z2$="A3*V*V": Z3$="B3*V": Z4$="C3"
372PRINT "DIFFERENTIAL ENERGY LOSS(ERGS/CM)"
373PRINTUSING 374,Z1$,Z2$,Z3$,Z4$
374%#########      ######          ####          ##
375GOSUB '01
376FOR I=1TO N1
378PRINTUSING 355,T1(I),A3*V1(I)*V1(I),B3*V1(I),C3
380NEXT I
382GOSUB '01
384FOR I=1TO 5
386PRINT
388NEXT I
389GOSUB '01
390PRINT "CUMULATIVE ENERGY LOSS(ERGS)"
391Z2$="INT(A3*V*V)DS"
392Z3$="INT(B3*V)DS"
393Z4$="INT(C3)DS"
396PRINTUSING 398,Z1$,Z2$,Z3$,Z4$
398%#########      #############  ###########    #########
399GOSUB '01
400Z6=T1(1)
402Z7=0.5*A3*(V0!2+V1(1)!2)*S1(1)*M
404Z8=0.5*B3*(V0+V1(1))*S1(1)*M
406Z9=C3*S1(1)*M
408PRINTUSING 355,Z6,Z7,Z8,Z9
410FOR I=2TO N1
412Z6=T1(I)
414Z7=Z7+0.5*A3*(V1(I-1)!2+V1(I)!2)*(S1(I)-S1(I-1))*M
416Z8=Z8+0.5*B3*(V1(I-1)+V1(I))*(S1(I)-S1(I-1))*M
418Z9=Z9+C3*(S1(I)-S1(I-1))*M
420PRINTUSING 355,Z6,Z7,Z8,Z9
422NEXT I
424GOSUB '01
430END
5100DEFFN'11(T)
5102REM CALCULATES VELOCITY Q7=001
5105V=V0-C3*T
5110V1(I)=V
```

```
5115RETURN
5300DEFFN'13(T)
5302REM CALCULATES VELOCITY Q7=011
5305V=B3*T:V=EXP(-V)
5310V=V*(B3*V0+C3)
5315V=(V-C3)/B3
5320V1(I)=V
5325RETURN
5600DEFFN'16(T)
5602REM CALCULATES VELOCITY Q7=111
5605IF F1]1THEN 5645
5607REM  LOG FORM
5610R1=B3!2-4.0*A3*C3:R1=SQR(R1)
5615F3=2.0*A3*V0+B3+R1
5620F3=F3/(F3-2.0*R1)
5625V=EXP(R1*T): V=V*F3*(R1-B3)
5630V=V+B3+R1
5635 V=V/(2.0*A3*(F3*EXP(R1*T)-1.0))
5640GOTO 5694
5645IF F1]2THEN 5680
5647REM  TRIG FORM
5650R1=4.0*A3*C3-B3!2: R1=SQR(R1)
5655V=2.0*A3*V0+B3:V=V/R1
5660V=ARCTAN(V): V=V-R1*T/2.0
5665V=TAN(V)-B3/R1
5670V=0.5*R1*V/A3
5675GOTO 5694
5680IF F1]3THEN 5692
5682REM ALGEBRAIC FORM
5685V=2*A3*V0+B3
5687V=2*A3*(2+T*V)
5690V=(4*A3*V0-B3*T*(2*A3*V0+B3))/V
5692A9=1:GOTO 5696
5694V1(I)=V
5696RETURN
6100 DEFFN'21
6102REM CALCULATES PENETRATION Q7=001
6105S=V0*V0-V*V
6110S=0.5*S/C3
6115S1(I)=S
6120RETURN
6300DEFFN'23
6302REM CALCULATES PENETRATION Q7=011
6305S=B3*V0+C3
6310S=(B3*V+C3)/S:S=LOG(S)
6315S=C3*S/B3
6320S=(V0-V+S)/B3
6325S1(I)=S
6330RETURN
6600DEFFN'26
```

```
6602REM CALCULATES PENETRATION Q7=111
6625Z1=A3*V*V+B3*V+C3
6630Z1=(A3*V0*V0+B3*V0+C3)/Z1
6635Z1=LOG(Z1)
6637S=B3*T1(I)
6640S=0.5*(Z1-S)/A3
6740 S1(I)=S
6745RETURN
7000DEFFN'01
7002REM PRINTS DASHED LINE
7005Z9$="- - - - - - - - - "
7010PRINT Z9$,Z9$,Z9$,Z9$
7015RETURN
```

DISTRIBUTION LIST

No. of Copies	Organization
12	Commander Defense Documentation Center ATTN: DDC-TCA Cameron Station Alexandria, VA 22314
1	Director of Defense Research and Engineering (OSD) Washington, DC 20301
1	Director Weapons Systems Evaluation Gp Washington, DC 20305
1	Chairman Defense Science Board Washington, DC 20301
1	Commander US Army Materiel Development and Readiness Command ATTN: DRCDMA-ST 5001 Eisenhower Avenue Alexandria, VA 22333
1	Commander US Army Aviation Systems Command ATTN: DRSAV-E 12th and Spruce Streets St. Louis, MO 63166
1	Director US Army Air Mobility Research and Development Laboratory Ames Research Center Moffett Field, CA 94035
1	Commander US Army Electronics Command ATTN: DRSEL-RD Fort Monmouth, NJ 07703
5	Commander US Army Missile Command ATTN: DRSMI-R DRSMI-RDP, J.M. Brown DRSMI-RS, R. Petty DRCPM-RK DRCPM-MD Redstone Arsenal, AL 35809
1	Commander US Army Tank Automotive Development Command ATTN: DRDTA-RWL Warren, MI 48090
1	Commander US Army Mobility Equipment Research & Development Command ATTN: Tech Docu Cen, Bldg. 315 DRSME-RZT Fort Belvoir, VA 22060
1	Commander US Army Armament Command Rock Island, IL 61202
2	Commander US Army Frankford Arsenal ATTN: SARFA-J7400 SARFA-J7600 Philadelphia, PA 19137
3	Commander US Army Picatinny Arsenal ATTN: SARPA-AD-G-S, Gaydos SARPA-AD-L, S. Einbinder SARPA-AD-C, C.J. Killen Dover, NJ 07801
1	Commander US Army Watervliet Arsenal ATTN: SARWV-RDD-AT, R. Thierry Watervliet, NY 12189

DISTRIBUTION LIST

No. of Copies	Organization
1	Commander US Army Harry Diamond Labs ATTN: DRXDO-TI 2800 Powder Mill Road Adelphi, MD 20783
4	Commander US Army Materials and Mechanics Research Center ATTN: Dr. R. Frost Mr. A. Alesi Mr. K. Abbott Mr. E. DeLuca Watertown, MA 02172
6	Commander US Army Natick Research and Development Command ATTN: Mr. T. Bailey Mr. T. Keville Mr. W. Wright Mr. F. Figucia Mr. M. Donomee MAJ W.R. Belcher Natick, MA 01760
1	Director US Army TRADOC Systems Analysis Activity ATTN: ATAA-SA White Sands Missile Range NM 88002
1	Commander US Army Training and Doctrine Command ATTN: MAJ R. Young Fort Monroe, VA 23651
1	Commander US Army Armor School ATTN: Armor Agency Fort Knox, KY 40121

No. of Copies	Organization
1	Commander US Army Aviation School ATTN: Aviation Agency Fort Rucker, AL 36362
1	Commander US Army Field Artillery School ATTN: Field Artillery Agency Fort Sill, OK 73405
1	Commander US Army Infantry School ATTN: MAJ B. David Fort Benning, GA 31905
1	The Surgeon General Department of the Army Washington, DC 20310
1	HQDA (DAMO-CSS, MAJ Rogers) Washington, DC 20310
1	Chief of Naval Operations ATTN: OP-03EG Department of the Navy Washington, DC 20350
1	Commander US Naval Air Systems Command ATTN: AIR-604 Washington, DC 20360
2	Commander US Naval Weapons Center ATTN: Code 753 Code 12 China Lake, CA 93555
1	Commander US Naval Medical Field Research Laboratories Camp LeJuene, NC 28542

DISTRIBUTION LIST

No. of Copies	Organization
1	Commander US Naval Research Laboratories ATTN: Code 8434, W. Ferguson Washington, DC 20375
1	Commandant US Marine Corps Washington, DC 20380
1	Director, MCDEC ATTN: MAJ D. Crow Quantico, VA 22134
1	AFATL (DLRW) Eglin AFB, FL 32542

Aberdeen Proving Ground

Cdr, USAEA
  ATTN: SAREA-BL-B
Dir, USAMSAA
  ATTN: Dr. Sperrazza
Dir, USAHEL
  ATTN: Mr. B. Corona
Marine Corps Ln Ofc